Spring Time Galaxy Cluster Supernova Survey

Introduction

This is a *Mathematica* Notebook that will control a Celestron Nexstar telescope mount to slew to galaxies, capture images and upload them to a Wolfram Datadrop to be analysed for the prescence of a supernova.

The process for doing this is quite simple, first we will open a connection to the telescope via the serial port and define some functions for slewing. Then we will query the Wolfram knowledgebase to find galaxies above a certain magnitude. Once we have these two things we can begin to send slew commands to the telescope for each galaxy and capture an image. The image capture will happen via a USB video capture card with a Revolution Imager camera attached. We will wait a number of seconds for the mount to finish slewing and for the exposure to complete then we will run a command line script to capture the image and upload to a Wolfram Datadrop.

Telescope Connection

Mathematica/Wolfram provides some easy functions to communicate to devices via a serial port. The raspberry pi does not have a native serial port so we will be using a USB to Serial convertor which are easily obtainable. We will define some functions use DeviceOpen to open a serial connection and then some functions to create the command string that the Nexstar mount requires.

```
ScopeConnect[] := DeviceOpen["Serial", {"/dev/ttyUSB0", "BaudRate" -> 9600}];
```

```
TelescopeConnection = ScopeConnect[]
```

DeviceObject[Class: Serial ID: 1 Status:]

The following functions convert the RA/Dec values into the hex strings that the Nexstar protocol requires and generations the entire slew command.

```
RA2Hex[ra_Real] := ToUpperCase[IntegerString[Round[(ra / 360) * 65536], 16, 4]]
RA2Hex[ra_Quantity] := RA2Hex[FromDMS[ra]]
Declination2Hex[dec_Real] :=
 ToUpperCase[IntegerString[Round[(If[dec < 0, 360 + dec, dec] / 360) * 65536], 16, 4]]
Declination2Hex[dec_Quantity] := Declination2Hex[FromDMS[dec]]
```

```
SlewString[ra_, dec_] := "R" <> RA2Hex[ra] <> "," <> Declination2Hex[dec]
```

And now the actual Slew command that will send the command to the mount.

```
Slew[scope_, ra_, dec_] := Module[{},
 DeviceWrite[scope, SlewString[ra, dec]];
 ]
```

Data Drop

Our next step is establish our connection to Wolfram Data Drop - this is where we will store our images.
We're not creating a new databin here because ideally you'd want the images going to the same bin every night you run the survey, please replace the databin Id with yours, below!

```
DestinationDatabin = Databin["l4VXBScV"]
```

Databin[Short ID: l4VXBScV Entry count:]

Finding our Galaxies

Now we will obtain the list of galaxies we want to search for possible supernovae. We will use *Mathematica*'s GalaxyData function which queries the Wolfram Knowledgebase to obtain the list. We are interested right now in Galaxies that are part of the Virgo Galaxy Cluster, and we will further limit this to the top 50 brightest galaxies in the group that are above an altitude of 20 degrees.

please note that downloading data from wolfram knowledgebase can take a while so be patient!

```
VirgoGalaxies = GalaxyData[EntityClass["Galaxy", "VirgoCluster"],
   {"Name", "RightAscension", "Declination", "ApparentMagnitude", "Altitude"}];
```

```
AboveAltitudeVirgoGalaxies = Select[VirgoGalaxies, FromDMS[#[[5]]] > 20 &];
BrightestVirgoGalaxies = Take[SortBy[AboveAltitudeVirgoGalaxies, #[[4]] &], 50];
```

And lets take a look at the brightest galaxy in our list.

```
Take[BrightestVirgoGalaxies, 1]
```

```
{{M49, 12h 29m 46.7s , 8.0° , 8.3, 61 degrees 43 arc minutes 57. arc seconds}}
```

Image Acquisiton

Now we will look at the nuts and bolts around aquiring the image. Unfortunately *Mathematica* on the rasperry pi does not support capturing images from regular video capture devices, only the dedicated raspberry pi video board that is available. To get around this please install the “streamer” and “v4l-utils” packages via “sudo apt-get install streamer”

Please make sure the camera is installed in the telescope, focused and set to whatever settings give you the most sensitivity. For my skies/scope combination I would use the x128 integration time with 12db gain and DNR at 6

And we’ll define a function to capture an image from the device using streamer by capturing into a temporary file and then importing that file into *Mathematica*. Because the USB video capture devices can be a bit wonky with the first initial frames we ask streamer to pull 10 frames and we import the last one.

```
CaptureTelescopeImage[] := Module[{},
   Run["streamer -s 720x576 -t 10 -n pal -o /tmp/survey00.jpeg "];
   Import["/tmp/survey09.jpeg"]
  ];
```

And let’s try it out. If you see an image below we’re good to go:

```
CaptureTelescopeImage[]
```

In addition to capturing the image we should do some basic image processing before we place it into the data bin. We will apply a MaxFilter on the image which looks at the neighbouring pixels in an image and sets them to the local maximum, this will have the effect of enlarging stars and reducing background noise! We then apply a binary filter to the image with a threshold that will convert the image to pure black and white with the white pixels being those that were above 40% grey, and lastly we reduce the size of the image to 320 pixels horizontal. Doing all of this makes it easier to identify stars (and hopefully supernova) and reduces the size of the image so we can store more in the datadrop.

```
ProcessImage[image_] :=
  ImageResize[
   Binarize[
    MaxFilter[
     ColorConvert[image, "GrayScale"], 4],
    .4],
   320];
```

Putting it Together

We have all the pieces we need now to start the process. We'll start by defining a function

that given an entry from our list of galaxies will send the slew command, wait a predefined amount of time, capture and process the image

```
WaitTime = 60;
SlewAndCapture[entry_] :=
       Module[{},
        PrintTemporary["Slewing to: " <> entry[[1]]]
         Slew[TelescopeConnection, entry[[2]], entry[[3]]];
            PrintTemporary["Waiting"]
             Pause[WaitTime];
                PrintTemporary["Capturing: " <> entry[[1]]];
        <| "name" → entry[[1]],
           "date" → DateObject[],
           "stars" → ProcessImage[CaptureTelescopeImage[]] |>];
```

And now we process each entry in our list and post the data to the datadrop:

```
DatabinAdd[DestinationDatabin, SlewAndCapture[#]] & /@ BrightestVirgoGalaxies;
```

`